FONCTIONS D'UNE VARIABLE RÉELLE : DÉRIVABILITÉ

SOLUTIONS DES EXERCICES

Bernard Dupont

Bernard.Dupont@univ-lille1.fr

Exercice M.1 (dérivée logarithmique)

Enoncé

Soit f une fonction dont l'ensemble d'arrivée est inclus dans $]0;+\infty[$.

1. Donner sa dérivée logarithmique avec les opérateurs **diff** et **D**.

2. Soit $f(x) = x^a$. Donner sa dérivée logarithmique en $x = 2$. Quel opérateur vous semble le plus pratique?

Solution

1. Comme son nom l'indique, la dérivée logarithmique de f est la dérivée première de $ln f(x)$. Avec la commande **diff**, le premier argument est l'expression **log(f(x))**. Avec la commande **D**, l'argument est la composition de **log** et de **f**.

```
> restart;
  diff(log(f(x)),x);#dérivée logarithmique en un point
  D(log@f);#fonction dérivée logarithmique
```

$$\frac{\frac{d}{dx} f(x)}{f(x)}$$

$$\left(z \to \frac{1}{z}\right) @ f \, D(f) \tag{1.1}$$

2. On pose l'expression et la fonction.

```
> f(x):=x^alpha;
  f:=unapply(f(x),x);
```

$$f(x) := x^{\alpha}$$

$$f := x \to x^{\alpha} \tag{1.2}$$

L'opérateur **D** est plus simple à utiliser que **diff** car il demande moins de manipulations :

```
> diff(log(f(x)),x);eval(%,x=2);
  D(log@f)(2);
```

$$\frac{\alpha}{x}$$

$$\frac{1}{2}\alpha$$

$$\frac{1}{2}\alpha \tag{1.3}$$

Exercice E.1 (élasticités de demande et d'offre)

Enoncé

1. Soit f une fonction de la variable x : $y = f(x)$. Rappelez la définition de l'élasticité de y par rapport à x. Quelle est son interprétation économique?
2. La demande d'un produit dépend de son prix. Calculez l'élasticité-prix du bien quand la demande est une fonction affine puis quand elle est une fonction puissance. Dans chaque cas, représentez graphiquement la fonction de demande et l'élasticité prix correspondante.
3. L'offre d'un produit dépend de son prix. Calculez l'élasticité-prix du bien quand l'offre est une fonction affine puis quand elle est une fonction puissance. Dans chaque cas, représentez graphiquement la fonction d'offre et l'élasticité prix correspondante.

Solution

1. Soit f une fonction de la variable x : $y = f(x)$. Rappelez la définition de l'élasticité de y par rapport à x. Quelle est son interprétation économique?
2. La demande d'un produit dépend de son prix. Calculez l'élasticité-prix du bien quand la demande est une fonction affine puis quand elle est une fonction puissance. Dans chaque cas, représentez graphiquement la fonction de demande et l'élasticité prix correspondante.
3. Même question pour l'offre du produit.

Solution

1. En économie, x est une variable explicative et y est une variable expliquée. On a la relation "causale" $y = f(x)$. Notons $E(y, x)$ l'élasticité de y par rapport à x. Par définition :

$$E(y, x) = \lim_{h \to 0} \frac{\dfrac{f(x+h) - f(x)}{f(x)}}{\dfrac{(x+h) - x}{x}} = x\frac{f'(x)}{f(x)} = x\frac{d}{dx}(\log|f(x)|).$$

Elle s'interprète comme le pourcentage de variation de la variable expliquée consécutif à une variation de 1% de la variable explicative. Les deux opérateurs de dérivation **diff** et **D** conviennent pour calculer l'élasticité.

```
> restart;
  Eyx_1:=x*diff(log(f(x)),x);
  Eyx_2:=x*D(log@f)(x);
```

$$Eyx\_1 := \frac{x \left(\frac{d}{dx} f(x)\right)}{f(x)}$$

$$Eyx\_2 := \frac{x \, D(f)(x)}{f(x)} \qquad \textbf{(2.1)}$$

2. L'élasticité-prix de la demande est variable pour une fonction affine et constante pour une fonction puissance.

```
> Qd1:=P->-a*P+b;#fonction affine
  EQP1_1:=P*diff(log(Qd1(P)),P);#utilisation de diff
  EQP1_2:=P*D(log@Qd1)(P);#utilisation de D
  a,b:=1,10;
  plot({Qd1(P),EQP1_1},P=0...9);
```

$$Qd1 := P \to -a\,P + b$$

$$EQP1\_1 := -\frac{P}{-P+10}$$

$$EQP1\_2 := -\frac{P}{-P+10}$$

$$a, b := 1, 10$$




```
> Qd2:=P->A*P^(-a);#fonction puissance
  EQP2_1:=P*diff(log(Qd2(P)),P);
  EQP2_2:=P*D(log@Qd2)(P);
  A,a:=10,5;
  plot({Qd2(P),EQP2_1},P=1...3);
```

$$Qd2 := P \to A\,P^{-a}$$

$$EQP2\_1 := -5$$

$$EQP2\_2 := -5$$

$$A, a := 10, 5$$



3. L'élasticité-prix de l'offre est également variable pour une fonction affine et constante pour une fonction puissance.

```
> Qo1:=P->a*P+b;#fonction affine
  EQoP1_1:=P*diff(log(Qo1(P)),P);#utilisation de diff
  EQoP1_2:=P*D(log@Qo1)(P);#utilisation de D
  a,b:=1,-1;
  plot({Qo1(P),EQoP1_1},P=2...4);
```

$$Qo1 := P \rightarrow a\,P + b$$

$$EQoP1\_1 := \frac{P}{P-1}$$

$$EQoP1\_2 := \frac{P}{P-1}$$

$$a, b := 1, -1$$




```
> Qo2:=P->A*P^(a);#fonction puissance
  EQoP2_1:=P*diff(log(Qo2(P)),P);
  EQoP2_2:=P*D(log@Qo2)(P);
  A,a:=1,5;
  plot({Qo2(P),EQoP2_1},P=1...2.5);
```

$$Qo2 := P \rightarrow A\,P^{a}$$

$$EQoP2\_1 := 5$$

$$EQoP2\_2 := 5$$

$$A, a := 1, 5$$



Exercice E.2 (taux marginal de substitution)

Enoncé

Soit la fonction d'utilité continûment différentiable : $(x, y) \rightarrow U(x, y)$, où x et y désignent les quantités consommées de deux biens. Le niveau d'utilité étant fixé, quel est le taux d'échange psychologique entre les deux biens?

Solution

Fixons le niveau d'utilitéà U_0. On a $U(x, y) = U_0$ et les quantités consommées y sont implicitement fonction des quantités consommées x. Par le théorème de la fonction implicite, on connaît l'impact d'une (petite) augmentation de x sur y. On utilise la commande `implicitdiff` pour déterminer le taux marginal de substitution.

```
> restart;
  implicitdiff(U(x,y)-Uo,y(x),x);
  convert(%,diff);
```

$$-\frac{D_1(U)(x, y)}{D_2(U)(x, y)}$$

$$-\frac{\frac{\partial}{\partial x} U(x, y)}{\frac{\partial}{\partial y} U(x, y)} \tag{3.1}$$

Le taux marginal de substitution est l'opposé du rapport des utilités marginales. Comme celles-ci sont positives, on en déduit qu'à niveau de satisfaction donné, l'augmentation des quantites consommées x est compensée par une diminution des quantités consommées y.

Exercice E.3 (taux marginal de substitution technique)

Enoncé

Soit la fonction de production générale : $(l, k) \rightarrow F(l, k)$, où et désignent les quantités de travail et de capital utilisées. Le niveau de production étant fixé, Comment faut-il compenser la (petite) baisse d'un facteur pour maintenir constant le volume de production?

Solution

Si le niveau de production est fixé à Y_0, et sous hypothèse de substituabilité des deux facteurs de production, la moindre utilisation d'un facteur doit entrainer l'augmentation d'utilisation de l'autre facteur. Le taux de substitution technique s'obtient par application du théorème de la fonction implicite à l'équation $F(l, k) = Y_0$.

```
> implicitdiff(F(l,k)-Yo,k(l),l);convert(%,diff);
```

$$-\frac{D_1(F)(l, k)}{D_2(F)(l, k)}$$

$$-\frac{\frac{\partial}{\partial l} F(l, k)}{\frac{\partial}{\partial k} F(l, k)} \qquad \textbf{(4.1)}$$

Le taux de substitution technique est l'opposé du rapport des productivités marginales, qui sont toutes deux des quantités positives. On compense donc la baisse d'un facteur par une hausse de l'autre.